
Abstract. We illustrate the use of logic programming techniques for
finite model checking of CTL formulae. We present a technique for infinite state model checking of safety properties based upon logic program
specialisation and analysis techniques. The power of the approach is illustrated on several examples. For that, the efficient tools logen and
ecce are used. We discuss how this approach has to be extended to handle more complicated infinite state systems and to handle arbitrary CTL
formulae.

Introduction

Recent years have seen dramatic growth [9] in the application of model checking [8, 5] techniques to the validation and verification of correctness properties of
hardware, and more recently software systems. One of the methods is to model
a hardware or software system as a finite, labeled transition system (LTS) which
is then exhaustively explored to decide whether a given temporal specification
holds. Recently, there has been increasing interest in applying logic programming techniques to model checking. Table-based logic programming can be used
as an efficient means of performing explicit model checking [35] and set-based
logic program analysis for model checking is explored in [7].
However, despite the success of model checking, most systems must be substantially simplified (i.e., abstracted) and “considerable human guidance and ingenuity is generally required to transform the original problem to a form where
the final push button automation can be applied” [36]. Furthermore, most software systems can conveniently be modelled by infinite state systems: as soon as
some kind of recursion, dynamic or unbounded data structures come into play,
an unbounded number of states can be reached and must be verified. In practice
the number of possible states in implementations is always finite but may be
so huge as to make any exhaustive approach futile. This probably explains why,
contrary to the situation in hardware, verification in general and model checking
in particular has had hardly any impact on standard software practice.
For these reasons, there has recently been considerable interest in infinite
model checking (e.g., [3, 39, 32, 12, 2]). This, by its generally undecidable nature,
is a daunting task, for which abstraction is the key issue [9]. Indeed, abstraction

allows one to approximate an infinite system (or a complicated finite one) by a
(simple) finite one, and if proper care is taken the results obtained for the finite
abstraction will be valid for the infinite system.
This research aims at exploring automatic means of building precise but
tractable abstractions for infinite model checking (or model checking of finite,
but very complex systems). We propose to do this by extending technology that
has been developed to tackle similar problems in the context of automatic logic
program analysis and specialisation. In essence, we:
– model a system to be verified as a logic program. This obviously includes
finite LTS, but also allows to express systems with an infinite number of
states. Note that this translations is often very straightforward, due to the
built-in support of logic programming for non-determinism and unification.
– model the full temporal logic CTL as a logic program interpreter acting on
the representation above. This interpreter will make use of the tight link
that exists between the semantics of logic programs and least-fixed points.
– and then try to automatically derive abstractions for infinite model checking
through a combination of partial evaluation and abstract interpretation technology. The tools logen [19, 26] and ecce [25, 27] are used successfully to
automate the work.
The contribution of the paper is the development of a correct CTL interpreter
in logic programming and its use as a sound basis for model checking of finite
and infinite state systems via program analysis and specialisation in general and
the tools logen and ecce in particular. This paper builds upon the initial insights and experiments in [14] where it was shown that abstraction-based partial
deduction can be used as a powerful inversion tool.
In the following, we present the logic CTL [11] to express the properties we
want to verify and its translation into a logic program. We discuss the validity of
model checking infinite systems. We implement as logic programs, systems expressed as labeled transition systems, Petri nets, processes synchronised through
shared variables, etc. We then show how existing technology for the specialisation and analysis of logic programs can be used to achieve some model checking
tasks of infinite state systems. We present some successful experiments, but also
shortcomings of the existing systems. We then give directions for further research
to enable model checking of arbitrary CTL formulae on infinite state systems.

2
2.1

The model-checking of CTL in logic programming
CTL syntax and semantics

The temporal logic CTL (Computation Tree Logic) introduced by Clarke and
Emerson in [11], allows to specify properties of specifications generally described
as Kripke structures. The syntax and semantics for CTL are given below.
Given Prop, the set of propositions, the set of CTL formulae φ is inductively
defined by the following grammar (where p ∈ Prop):
φ := true | p |¬φ | φ ∧ φ | ∀ ° φ | ∃ ° φ | ∀φUφ | ∃φUφ

A CTL formula φ can be either true or false in a given state. For example,
true is true in all states, ¬true is false in all states, and p is true in all states which
contain the elementary proposition p. The symbol ° is the nexttime operator
and U stands for until. ∀ ° φ (resp. ∃ ° φ) intuitively means that φ holds
in every (resp. some) immediate successor of the current program state. The
formula ∀φ1 Uφ2 (resp. ∃φ1 Uφ2 ) intuitively means that for every (resp. some)
computation path, there exists an initial prefix of the path such that φ 2 holds
at the last state of the prefix and φ1 holds at all other states along the prefix.
The semantics of CTL formulae is defined with respect to a Kripke structure
(S, R, µ, s0 ) with S being the set of states, R(⊆ S × S) the transition relation,
µ(S → 2P rop ) giving the propositions which are true in each state, and s0 being
the initial state. Figure 1a gives a graphical representation of a Kripke structure
with 3 states s0 , s1 , s2 , where s0 is the initial state. The propositions p, q, r
“label” the states. Generally it is required that any state has at least one

Fig. 1. Example of a Kripke structure.

outgoing vertex. From a Kripke structure, we can define an infinite transition
tree as follows: the root of the tree is labelled by s0 . Any vertex labelled by s has
one son labelled by s0 for each vertex s0 with a transition s → s0 in the Kripke
structure. For instance, the Kripke structure of Fig. 1a gives the prefix for the
transition tree starting from s0 given in Fig. 1b.
If the system is not directly specified by a Kripke structure but e.g. by a
Petri net, the markings and transitions between them give resp. the states and
transition relation of the Kripke structure.
The CTL semantics is defined on states s by:
– s |= true
– s |= p iff p ∈ P (s)
– s |= ¬φ iff not(s |= φ)
– s |= φ1 ∧ φ2 iff s |= φ1 and s |= φ2
– s |= ∀ ° φ iff for all state t such that (s, t) ∈ R, t |= φ
– s |= ∃ ° φ iff for some state t such that (s, t) ∈ R, t |= φ
– s |= ∀φ1 Uφ2 iff for all path (s0 , s1 , . . .) of the system with s = s0 , ∃i(i ≥
0 ∧ si |= φ2 ∧ ∀j(0 ≤ j < i → sj |= φ1 ))
– s |= ∃φ1 Uφ2 iff it exists a path (s0 , s1 , . . .) with s = s0 , such that ∃i(i ≥
0 ∧ si |= φ2 ∧ ∀j(0 ≤ j < i → sj |= φ1 ))

If we look at the unfolding of the Kripke structure, starting from s0 (Fig. 1b),
we can see, for instance, that
– s0 |= ∃ ° p since there exists a successor of s0 (i.e. s2 ) where p is true
(p ∈ µ(s2 )),
– s0 6|= ∀ ° p since in some successor of s0 , p does not hold.
– s0 |= ∀pUq since the paths from s0 either go to s1 where q is true or to s2
and then directly to s1 . In both cases, the paths go through states where p
is true before reaching one state where q holds.
Since they are often used, the following abbreviations are defined
– ∀3φ ≡ ∀ true Uφ
i.e. for all paths, φ eventually holds,
– ∃3φ ≡ ∃ true Uφ
i.e. there exists a path where φ eventually holds,
– ∃2φ ≡ ¬∀3(¬φ)
i.e. there exists a path where φ always holds,
– ∀2φ ≡ ¬∃3(¬φ)
i.e. for all paths φ always holds.
E.g. ∀2φ states that φ is an invariant of the system, ∀3φ states that φ is
unavoidable and ∃3φ states that φ may occur.
2.2

CTL as fixed point and its translation to logic programming

One starting point of this paper is that both the Kripke structure defining the
system under consideration and the CTL specification can be easily defined using
logic programs. This is obvious for the standard logic operators as well as for
the CTL operators ∃°. The following equality can be easily proved ∀ ° φ ≡
¬∃°¬φ. Moreover, the operators ∃U and ∀U can be defined as fixpoint solutions
[31]. Indeed, if you take the view that φ represents the set of states S where φ is
valid, it can be proved that ∃pUq ≡ µY = (q ∨ (p ∧ ∃ ° Y )) where µ stands for
the least fixpoint of the equation. Intuitively this equation tells that the set of
states satisfying ∃pUq is the least one satisfying q or having a successor which
satisfies this property.
∃3 and ∀2 can be derived from what precedes. ∀° can be derived using the
equivalence ∀ ° φ ≡ ¬∃ ° ¬φ. Slightly more involved is the definition of the
set of states which satisfy ∃2φ. These states can be expressed as the solution
of νX = φ ∧ ∃ ° X where ν stands for the greatest fixpoint of the equation.
Greatest fixpoint cannot be directly computed by logic programming systems,
but the equation can be translated into ∃2φ ≡ ¬Φ where the set Y of states
which satisfy Φ is defined by µY = ¬φ ∨ ¬∃ ° ¬Y . Finally, ∀3φ ≡ ¬∃2¬φ
and ∀φ1 Uφ2 ≡ ¬(∃¬φ2 U(¬φ1 ∧ ¬φ2 )) ∧ ¬∃2¬φ2 . This last equality says that
a state satisfies ∀φ1 Uφ2 if it has no path where φ2 is continuously false until a
state where both φ1 and φ2 are false nor a path where φ2 is continuously false.
We can see that we only use least fixpoint of monotonic equations. Moreover,
the use of table-based Prolog (XSB Prolog) ensures proper handling of cycles.
For infinite state systems, the derivation of the fixpoints cannot generally be
completed. However, by the monotonicity of all the used fixpoints, all derived
implications belong indeed to the solution (no overapproximation).
Our CTL specification is independant of any model. It only supposes that the
successors of a state s can be computed (through the predicate trans) and that

the elementary proposition of any state s can be determined (through prop). In
Fig. 2 we present a particular implementation of CTL as a (tabled) logic program.
For example, it can be used to verify the mutual exclusion example from [8],
simply by running it in XSB-Prolog. This follows similar lines as [35, 28] where
tabled logic programming is used as an (efficient) means of finite model checking.
Nonetheless, our translation of CTL in this paper is expressed (more clearly) as a
meta-interpreter and will be the starting point for the model checking of infinite
state systems using program specialisation and analysis techniques.
/* A Model Checker for CTL fomulas written for XSB-Prolog */
sat(_E,true).
sat(_E,false) :- fail.
sat(E,p(P)) :- prop(E,P). /* elementary proposition */
sat(E,and(F,G)) :- sat(E,F), sat(E,G).
sat(E,or(F,_G)) :- sat(E,F).
sat(E,or(_F,G)) :- sat(E,G).
sat(E,not(F)) :- not(sat(E,F)).
sat(E,en(F)) :- trans(_Act,E,E2),sat(E2,F). /* exists next */
sat(E,an(F)) :- not(sat(E,en(not(F)))). /* always next */
sat(E,eu(F,G)) :- sat_eu(E,F,G).
/* exists until */
sat(E,au(F,G)) :- sat(E,not(eu(not(G),and(not(F),not(G))))),
sat_noteg(E,not(G)). /* always until */
sat(E,ef(F)) :- sat(E,eu(true,F)). /* exists future */
sat(E,af(F)) :- sat_noteg(E,not(F)). /* always future */
sat(E,eg(F)) :- not(sat_noteg(E,F)).
/* exists global */
/* we want gfp -> negate lfp of negation */
sat(E,ag(F)) :- sat(E,not(ef(not(F)))). /* always global */
:- table sat_eu/3. /* table to compute least-fixed point using XSB */
sat_eu(E,_F,G) :- sat(E,G). /* exists until */
sat_eu(E,F,G) :- sat(E,F), trans(_Act,E,E2), sat_eu(E2,F,G).
:- table sat_noteg/2. /* table to compute least-fixed point using XSB */
sat_noteg(E,F) :- sat(E,not(F)).
sat_noteg(E,F) :- not((trans(_Act,E,E2),not(sat_noteg(E2,F)))).
Fig. 2. CTL interpreter

Model checking of infinite systems The infinite state system we handle are
finitely branching. In fact, what is required is to be able, in a finite number of
“steps”, to look at all the succesors of a state. (In real-time systems a state can
have an infinite number of direct successors, and model checking then requires
more sophisticated symbolic methods.) We have to show that, using these infinite
state systems, our CTL interpreter is correct wrt. the SLS-semantics [34]. This
is because our analysis and specialisation might replace infinite failure by finite
failure but do so only in accordance with the SLS-semantics.
Indeed, the only potential loops are linked with ∃U or ∀U. In fact, we can
notice that the computation is based on two fixpoint calculations; one is defined
through sat eu and the other through sat noteg. If we look at sat eu (the

treatment of sat noteg will pose similar problems), three cases may occur in
the SLD-resolution:
– A path is found satisfying F Until G (success; no difference with SLDNFsemantics).
– It is found that no path satisfies F Until G, i.e. all the paths satisfy ¬GU(¬F ∨
¬G) (finite failure; no difference with SLDNF-semantics)
– The resolution loops in an infinite path satisfying F . In this case the system
will not reply if no means is given to detect this infinite path. However, wrt
the SLS-semantics the answer is no (which is different from the SLDNFsemantics), which is the correct answer according to the CTL-semantics we
have given earlier (no path satisfies the requested property).

3

The systems analysed

We illustrate our approach by analysing finite or simple but infinite states systems specified initially by LTS, Petri nets or parallel processes using shared
variables. This section presents these systems.
3.1

Petri net of a simple mutual exclusion problem

Fig. 3. Petri net with a single semaphore

Figure 3 models a single process which may enter a critical section (cs), the
access to which is controlled by a semaphore (sema). This Petri net can be
encoded directly as trans/3 facts for our CTL interpreter:
trans(enter_cs,[s(X),s(Sema),CritSec,Y,C],[X,Sema,s(CritSec),Y,C]).
trans(exit_cs, [X,Sema,s(CritSec),Y,C],[X,s(Sema),CritSec,s(Y),C]).
trans(restart,[X,Sema,CritSec,s(Y),C],[s(X),Sema,CritSec,Y,s(C)]).

3.2

Petri net of a manufacturing system

The manufacturing system in Fig. 4, used in [3], has been analysed. It models
an automated manufacturing system with 4 machines, 2 robots, 2 buffers (x 10
and x15 ) and an assembly cell. The initial marking is such that x1 = p for some
nonnegative parameter p. In [3], Bérard and Fribourg have used Hytech [17] to
discover a potential deadlock when p > 8.
3.3

Producer/consumer processes with shared variables

The following is a simplified version of the producer/consumer example from [1],
using a buffer of length 1. A state of the system is not only, as in the previous

cases, a tuple of natural numbers but this time contains items (simple values)
and lists of items. A “state” of the system is given by [A,Prod,In,Out,Buf,CR,B]
where A is the list of items which remain to be produced, Prod the last item
which has been produced, In the number of empty places in the buffer, Out the
number of full places in the buffer, Buff the buffer and CR the last item which
has been removed by the consumer. Finally B allows to check the items which
remain to be consumed (and is initially a copy of A).
trans(prod,[[X|T],prod,In,Out,Buf,CR,B], [T,add(X),In,Out,Buf,CR,B]).
trans(add, [A,add(X),1,0,Buf,CR,B], [A,prod,0,1,X,CR,B]).
trans(rem, [A,PA,0,1,Buf,cons,B], [A,PA,1,0,Buf,rem(Buf),B]).
trans(cons,[A,PA,In,Out,Buf,rem(Buf),[Buf|B]], [A,PA,In,Out,Buf,cons,B]).
trans(err ,[A,PA,In,Out,Buf,rem(Buf),[B2|B]],
[err,err,err,err,err,err,err]) :- Buf\==B2.

Fig. 4. Petri net representation of an automated manufacturing system with four machines, two robots, two buffers (x10 , x15 ) and an assembly cell.

4

Infinite Model Checking of Safety Properties

Before attempting to verify any CTL formula, let us restrict our attention to
checking safety properties, i.e., formulae of the form s |= ∀2saf e. Model checking
of such properties amounts to showing that there exists no trace which leads to
an invalid state, i.e., exploiting the fact that ∀2saf e ≡ ¬∃3(¬saf e).
Consider the Petri net in subsection 3.1. We first have to specify properties
of interest by defining prop/2: prop([X,Sema,s(s(CritSec)),Y,C],unsafe).
Here, we have specified that a state is unsafe if two or more processes are in
their critical section at the same time. Now, to check whether the above Petri net
can reach an unsafe state for an initial marking with 1 token in the semaphore

(sema), 0 tokens in the reset counter (c), no processes in the critical section (cs),
no processes in the final place (y) and X processes in the initial place (x) we
simply run the query: sat([X,0,s(0),0,0],ef(p(unsafe))).
Unfortunately, this query does not terminate under Prolog or XSB-Prolog
although the system does indeed satisfy the safety property for any value of X.
Even if we try to prove the property just for a particular value of X (e.g., X =
s(0)) neither Prolog nor XSB-Prolog [37] will terminate (even when adding moding or delay declarations). Indeed, the queries have infinitely failed SLD/SLGtrees with an infinite number of distinct call patterns (due to the counter c).
Thus, according to the well-founded semantics, we have that the program indeed entails not(sat([X,0,s(0),0,0],ef(p(unsafe)))), but existing procedures are unable to establish this.
In the following, we will be able to prove this safety property by a semanticspreserving program specialisation and analysis technique, specialising the CTL
interpreter for the query sat([X,0,s(0),0,0],ef(p(unsafe))) to the empty
program (in a logically sound fashion). For this we will proceed in three phases:
1. specialise the full CTL interpreter for the particular property using an offline
specialiser, so as to get rid of unneccesary complexities.
2. specialise the so obtained simplified interpreter using a full-fledged online
specialiser so as to obtain a finite, and as precise as possible abstraction of
the infinite state system for the property at hand.
3. use abstract interpretation to analyse this finite representation and determine whether the property is true, false, or undecided.
In the following we will describe and illustrate each of these phases.
4.1

Pre-compilation with logen

The complete CTL interpreter of Fig. 2 is quite complex, and makes heavy usage
of negation. The latter is difficult to handle by most current program analysis
and specialisation tools. However, for the particular class of formulae we want to
handle in this section we can get rid of the unnecessary complexity of full CTL by
pre-compiling the interpreter for the formula to be checked. This task is best performed by an offline specialiser, such as the logen system [19, 26]. For instance,
specialising Fig. 2 for the query sat([Processes,0,s(0),0,0],ef(p(unsafe)))
yields the much simplified interpreter below, which contains no negation and
where the Petri net has been compiled into the interpreter:
/* benchmark info: 1.17 ms */
/* atom specialised: sat([_264,0,s(0),0,0],ef(p(unsafe))) */
sat__0(B) :- sat_eu__1(B).
sat_eu__1([B,C,s(s(D)),E,F]).
sat_eu__1([s(G),s(H),I,J,K]) :- sat_eu__1([G,H,s(I),J,K]).
sat_eu__1([L,M,s(N),O,P]) :- sat_eu__1([L,s(M),N,s(O),P]).
sat_eu__1([Q,R,S,s(T),U]) :- sat_eu__1([s(Q),R,S,T,s(U)]).

This interpreter should now be called using the predicate sat 0(P) (which
corresponds to calling sat([P,0,s(0),0,0],ef(p(unsafe))) in the original

program). Observe that logen (and ecce as well) concatenates two underscores
and a unique identifier to existing predicate names.
The exact details of this compilation phase are not relevant for the present
article; all we need to know is that it terminates (actually it is also very efficient) and that it is totally correct in the sense that it preserves the computed
answers and finite failure. In other words, the specialised program succeeds for
a specialised query (e.g., sat 0(s(X))) with a computed answer θ (respectively
finitely fails) iff the original program does so for the corresponding original query
(e.g., sat([s(X),0,s(0),0,0],ef(p(unsafe))).
Observe that even this much simplified interpreter still does not terminate
under either SLD or tabled SLG resolution. Below, we will show how this interpreter can be analysed using automatic techniques and how we can show that
the encoded, infinite state Petri net obeys the safety property.
4.2

Online partial deduction with ecce

The second phase of our model checking technique will perform online partial
deduction using the ecce tool. It will construct a finite representation of the
infinite state space in the form of a specialised program. Below we present the
essential details of partial deduction and the essential details of the algorithm
used by ecce.
The underlying technique of partial deduction is to construct finite but possibly incomplete SLD(NF)-trees (i.e. a SLD(NF)-tree which, in addition to success
and failure leaves, may also contain leaves where no literal has been selected for
a further derivation step). These incomplete SLD(NF)-trees are obtained by applying an unfolding rule, defined as follows.
Definition 1. An unfolding rule is a function which, given a program P and a
goal G, returns a finite non-trivial1 and possibly incomplete SLD(NF)-tree τ for
P ∪ {G}. We also define leaves(τ ) to be the atoms in the leaf goals of τ .
Formally, the resultant of a branch of τ leading from the root G to a leaf goal
Gi via computed answer θ is the formula Gθ ← Gi . Partial deduction uses the
resultants for a given set of atoms S to construct the specialised program (and
for each atom in S a different specialised predicate definition will be generated).
Under the conditions stated in [29], namely closedness (all leaves are an instance
of an atom in S) and independence (no two atoms in S have a common instance),
total correctness of the specialised program is guaranteed (i.e., as above we
preserve the computed answers and finite failure).
We now present a concrete partial deduction algorithm based upon [27]. This
algorithm structures the set S of atoms to be specialised as a global tree γ: i.e.,
a tree whose nodes are labeled by atoms and where A is a descendant of B if
specialising label(B) lead to the specialisation of label(A). It outputs a set of
atoms A which can be used to construct a totally correct specialisation of P
1

A trivial SLD(NF)-tree is one in which no literal in the root has been selected for
resolution. Such trees are disallowed to obtain correct partial deductions.

for all instances of ← Q (possibly using a renaming transformation to ensure
independence).
Note that the algorithm below can be seen as a special kind of forwards
abstract interpretation (see [24]), where each atom in γ actually denotes all its
instances (i.e., the concretisations γ(A) of an atom A are all the instances of A).
To ensure termination of our algorithm, we have to ensure that the unfolding
rule U builds a finite SLD(NF)-tree. In addition, we have to guarantee that no
infinite branches are built up in the global tree γ: If it looks like an infinite
branch is being built up we have to abstract some of the atoms and restart the
process. We thus also have to ensure that this abstraction process itself cannot
be repeated infinitely often.
The following auxiliary concepts will help us to achieve this feat. First, to be
able to perform a suitable generalisation we define:
Definition 2. The most specific generalisation of a finite set of expressions S,
also denoted by msg(S), is the most specific expression M such that all expressions in S are instances of M .
Algorithms for calculating the msg exist [21], and we have for example
msg({p(0, s(0)), p(0, s(s(0)))}) = p(0, s(X)). We also have the important property, that for every expression A, there are no infinite chains of strictly more
general expressions. Now, to detect infinite branches, both in the global tree γ
and the SLD(NF)-trees constructed by U , we will use the homeomorphic embedding relation derived from [18, 20]. The following is the definition from [38]:
Definition 3. The (pure) homeomorphic embedding relation £ on expressions
is inductively defined as follows (i.e. £ is the least relation satisfying the rules):
1. X £ Y for all variables X, Y
2. s £ f (t1 , . . . , tn ) if s £ ti for some i
3. f (s1 , . . . , sn ) £ f (t1 , . . . , tn ) if ∀i ∈ {1, . . . , n} : si £ ti .
(Notice that n is allowed to be 0 and we thus have c £ c for all constant and
proposition symbols). The intuition behind the above definition is that A £ B
iff A can be obtained from B by “striking out” certain parts, or said another
way, the structure of A reappears within B. We have the important property
([18, 20]) that £ is a so-called well-quasi order on the set of expressions over
a finite alphabet, i.e., for every infinite sequence s1 , s2 , ... of expressions there
exists i < j such that si £ sj .
Algorithm 4.1 (partial deduction algorithm)
Input: a program P and a goal ← Q
Output: a set of atoms A and a global tree γ
Initialisation: γ := a “global” tree with a single unmarked node, labelled by Q
repeat
pick an unmarked leaf node L in γ
if ∃ a marked variantof L in γ then mark L
else if ∃ ancestor W of L such that label(W ) £ label(L) then
label(W ) := msg(L, W )

remove all descendants of W and unmark W
else
mark L
for all A ∈ leaves(U (P, label(L))) do
add a new unmarked C child of L to γ
label(C) := A
label(L → C) := a characteristic path i.e. a sequence of clauses in P which were
resolved with
until all nodes are marked
output A := {label(A) | A ∈ γ}

In this algorithm, M is a variant of L iff for some θ1 , θ2 : M θ1 = L and
Lθ2 = M . The notion of characteristic path is developped in [13, 27].
The above algorithm is parametrised by an unfolding rule U . Upon termination of the algorithm the closedness condition of [29] is satisfied, i.e., it is ensured
that together the atoms A with their SLD(NF)-trees form a complete description
of all possible computations that can occur for all concrete instances ← Aθ of
the goal of interest.
Note that ecce can also handle entire conjunctions of atoms [10] (instead of
just single atoms), but this was not required for the experiments in this paper.
By default, its control is also more refined in that it uses an extended £ [27, 22]
and characteristic trees on top of syntactic structure to control abstraction [27].
4.3

Most Specific Version Abstract Interpretation

The task of the abstract interpretation phase will be to do the verification proper
and try to infer whether the finite representation produced by the previous phase
admits a solution. For this we will use an abstract interpretation method based
on [30] which calculates so-called most specific versions of programs.
By mgu ∗ (A, B) we denote a particular idempotent and relevant most general
unifier of A and some B 0 , obtained from B by renaming apart wrt A (i.e., so
that B 0 and A have no variables in common). We also define the predicatewise application msg ∗ of the msg: msg ∗ (S) = {msg(Sp ) | p ∈ P red(P )}, where
Sp are all the atoms of S having p as predicate and P red(P ) denotes the set of
predicates occurring in the program P . In the following we define the well-known
non-ground TP operator (whose least fixed point gives the S-semantics [4]) along
with an abstraction TPα of it.
Definition 4. For a definite logic program P and a set of atoms A we define:
TP (A) = {Hθ1 . . . θn | H ← B1 , . . . , Bn ∈ P ∧ θi = mgu ∗ (Bi θ1 . . . θi−1 , Ai ) with
Ai ∈ A}. We also define TPα (A) = msg ∗ (TP (A)).
One of the abstract interpretation methods of [30] can be seen as calculating
lf p(TPα ) = TPα ↑∞ (∅) (this in turn can be seen as an abstract interpretation
method which infers top level functors for every predicate). The idea is to initially
proceed like TP , but if we get two or more success patterns for the same predicate
then we retain only one success pattern which covers them. TPα ↑∞ (∅) will

always stabilise after a finite number of iterations and it will produce a safe
approximation of the success set (i.e., any call which does not unify using mgu ∗
with an element of TPα ↑∞ (∅) will fail [finitely or infinitely]). In [30] more
specific versions of clauses and programs are obtained in the following way (which
preserves the least Herbrand model and the computed answers, but may replace
infinite by finite failure):
Definition 5. Let C = H ← B1 , . . . , Bn be a definite clause and A a set of
atoms. We define: msvA (C) = {Cθ1 . . . θn | θi = mgu ∗ (Bi θ1 . . . θi−1 , Ai ) with Ai ∈
A}. The more specific version msv(P ) of a program P is then obtained by replacing every clause C ∈ P by msvlf p(TPα ) (C) (note that msvlf p(TPα ) (C) contains at
most 1 clause).
Notably, the most specific version of a program without facts is the empty
program! Also observe that the above described analysis works backwards (or
bottom-up) from the facts to the query. It is thus an ideal complement to the
forwards analysis that partial deduction performs. (Ideally one would like to
perform the forwards and backwards analysis together [25]; generic algorithms
for this exist [25, 24] but they are not yet implemented.)
4.4

Putting it all together

Let us now return to checking the earlier mentioned safety property ¬∃3(unsaf e)
of the Petri net of Fig. 3, meaning that it is impossible to reach a marking where
two processes are in their critical section at the same time.
We have already compiled the formula ∃3(unsaf e) and our particular Petri
net into the interpreter by logen in Section 4.1.
Let us first attempt to prove that property for 2 processes. We thus apply
ecce to the compiled interpreter, specialising it for sat 0([s(s(0)),0,s(0),0,0])
(i.e., an initial marking with 2 processes and 1 token in the semaphore). This
yields the following specialised program (after a transformation time of 0.5 s):
sat__0([s(s(0)),0,s(0),0,0]) :- sat__0__1.
sat__0__1 :- sat_eu__1__2.
sat__0__1 :- sat_eu__1__3.
sat_eu__1__2 :- sat_eu__1__7(s(s(0)),s(0)).
sat_eu__1__2 :- sat_eu__1__3.
sat_eu__1__3 :- sat_eu__1__4.
sat_eu__1__4 :- sat_eu__1__5(s(s(0)),s(0)).
sat_eu__1__4 :- sat_eu__1__3.
sat_eu__1__5(A,s(B)) :- sat_eu__1__6(A,B).
sat_eu__1__5(s(A),B) :- sat_eu__1__5(A,s(B)).
sat_eu__1__6(A,B) :- sat_eu__1__5(s(A),B).
sat_eu__1__6(s(A),B) :- sat_eu__1__6(A,s(B)).
sat_eu__1__7(A,s(B)) :- sat_eu__1__8(A,B).
sat_eu__1__7(s(A),B) :- sat_eu__1__5(A,s(B)).
sat_eu__1__8(A,B) :- sat_eu__1__7(s(A),B).
sat_eu__1__8(s(A),B) :- sat_eu__1__8(A,s(B)).

As you can see ecce always generates one clause (the first one, defining
sat 0) which allows the specialised program to be used in the same way as the
original one and then clauses for a renamed (two underscores and the number 1
are usually added to the predicate name) and filtered (only variables occuring
in the query are left as arguments) version of the query (the clauses defining
sat 0 1). As this program contains no facts, the most specific version transformation trivially produces:
sat 0([s(s(0)),0,s(0),0,0]) :- fail.

This establishes the safety property: ∃3(unsaf e) is false and there is no way
that the system can reach a state where unsafe holds. As already mentioned,
this task cannot be established by Prolog or XSB-Prolog [37] with tabling.
Similarly, one can prove the safety property regardless of the number of processes, i.e., for any number of tokens in the initial place (x). When we specialise
the same compiled version of Fig. 2 for the query sat 0([X,0,s(0),0,0]) and
then compute the most specific version we get the following (in similar transformation times):
sat 0([X,0,s(0),0,0]) :- fail.

There are now of course two important questions that arise:
1. How can we be sure that the above implies the safety property? In fact, if
the safety property of the system in Fig. 3 is not satisfied, there must be
a trace of finite length leading to an unsafe state. Hence, by completeness
of SLD (and correctness of the CTL interpreter) we can deduce that there
should have been a computed answer for the query: sat 0([X,0,s(0),0,0]).
Now, as logen, ecce, and the most specific program technique all preserve
(provided there are no bugs in the implementations of course) failure and
the computed answers (cf. [29] and [30] respectively), and as the specialised
program fails we can conclude that the safety property does hold.
Indeed, the specialised program produced by any of the 3 phases is totally
correct: they do not remove any computed answer nor do they add any. So,
in a sense there is no over-approximation or under-approximation! Approximations only come into play if we analyse the residual programs.
For instance, we can produce the following safe over-approximation of the
success set: deduce that a call p(t̄) fails if it unifies with no clause in the
residual program (or just with a single clause p(s̄) ← f ail). For all other
calls deduce that they potentially succeed. Similarly, we can extract a safe
under-approximation of the success set: deduce that a call p(t̄) succeeds if
there is a fact p(s̄) ← in the residual program such that p(t̄) is an instance
of p(s̄). Otherwise deduce that the call potentially fails.
2. How did the system achieve this (automatically) ? In essence, the specialisation component (ecce) performed a symbolic traversal of the infinite state
space, thereby producing a finite representation of it, on which the analysis
component performed the verification of the specification. More precisely,
the following ingredients of our system seem to be relevant or even vital:
– the homeomorphic embedding £ ensures that we build a finite representation of the state space. At the same time £ is sufficiently powerful [23]

to minimise unnecessary abstraction, increasing the chances of successful
verification.
– characteristic trees ensure that we produce enough polyvariance to account for different behaviour of configurations. It further minimises the
risk of unnecessary, harmful abstraction (cf. [27]).
– abstract interpretation performs the model checking proper on the finite
representation obtained above.
– the abstract interpretation works backwards (from unsafe states towards
initial states) while the partial deduction works forwards (from initial
states to unsafe states). Our technique thus gives a combined backwards/forwards analysis (which should become even better by implementing the full integration of [25, 24]).
4.5

Tackling the manufacturing Petri net example

We applied the approach to the manufacturing system in subsection 3.2 and
were able to prove absence of deadlocks for parameter values of e.g., 1,2,3. When
leaving the parameter unspecified, the system did not establish an absence of
deadlocks and produced a (large) residual program containing facts. And indeed,
for parameter values ≥ 9 the system can actually deadlock. We are investigating whether a counter example can effectively be extracted from this residual
program. The runtimes of our system compare favourably with HyTech [3].

Coping with more complicated formalisms

In principle, it is possible to extend our approach to verify larger, more complicated infinite systems. (Notice that larger systems have been approached with
related techniques as a preprocessing phase [16]. However, their purpose is to
reduce the state space rather than provide novel ways of reasoning.) As with
all automatic specialisation tools, there are several points that need to be addressed: allow more generous unfolding and polyvariance (efficiency, both of the
specialisation process and the specialised program, are less of an issue in model
checking) to enable more precise residual programs and implement the full algorithm of [24] which allows for more fine grained abstraction and use BDD-like
representations whenever possible. We elaborate on some of these issues below.
Also, in theory, we can apply the power of our approach, to systems specified
in other formalisms such as the π-calculus (cf. the experiment in [15]) or processes
with synchronisations, simply by writing an interpreter for these formalisms in
logic programming.
Looking at the producer/consumer example presented in 3.3, we use the
same interpreter for Petri nets as in Fig. 3 but for more complex states. An
error occurs in the system of Subsection 3.3 when the next item consumed does
not correspond to the one expected. To encode this as well as the possible inital
states of our system, we add:

prop([err,err,err,err,err,err,err],unsafe).
err(A) :- sat([A,prod,1,0,vide,cons,A],ef(p(unsafe))).

Again, our 3-phased approach has achieved infinite model checking and has
inferred err(A) :- fail., i.e. for any list of items, the safety property is verified.
However, things are not always that easy and problems do appear with more
complicated systems.
Example 1. The following is a slightly more involved version of the produceconsume example from [1]. It still uses a buffer of length 1, but uses arithmetic
operations to test whether the buffer contains any item to consume. Here, In
(resp. Out) stands for the number of items which have been put in (resp. removed
from) the buffer. (In the full version of [1], we have tests of the form In <
Out + N , where N is the size of the buffer.)
trans(prod,[N,[X|T],prod,In,Out,Buf,CR,B], [N,T,add(X),In,Out,Buf,CR,B]).
trans(add, [N,A,add(X),In,Out,Buf,CR,B],
[N,A,prod,In1,Out,X,CR,B]) :- In < Out+1, In1 is In+1.
trans(rem, [N,A,PA,In,Out,Buf,cons,B],
[N,A,PA,In,Out1,Buf,rem(Buf),B]) :- In>Out, Out1 is Out+1.
trans(cons,[N,A,P,In,Out,Buf,rem(Buf),[Buf|B]],[N,A,P,In,Out,Buf,cons,B]).
trans(err ,[N,A,PA,In,Out,Buf,rem(Buf),[B2|B]],
[N,err,err,err,err,err,err,err]) :- Buf\==B2.

This example cannot be successfully verified by our current approach, due to
its inability to detect simple inconsistencies. For example, neither ecce nor the
technique of [30] will detect that the conjunction X < 1, X > 1 cannot succeed.
It should be possible to remedy this deficiency by adding CLP-techniques to
ecce and/or going to more sophisticated abstract domains as outlined in [24].
Similar extensions will probably be needed to handle real time systems [3, 17].
Other problems do appear when we move to formalisms where the state
representation gets more complex. For example, whereas for Petri nets a state
was just a sequence of natural numbers, in CCS a state is an (arbitrarily complex)
expression. As the following example shows, this leads to other difficulties.
Example 2. Take the following simple CCS specification of an agent P (where
“a” and “ā” are complementary actions, “.” denotes the action prefix, and “|”
the parallel composition):
P =Def a.P |ā.P
The transitional semantics of CCS tells us that the agent P can perform the
action a and ā respectively thanks to its left and right branch. Moreover, P
can perform the invisible action τ (via the synchronization of a and ā) leading
to a new expression P |P which in turn can perform τ leading to (P |P )|P (or
P |(P |P )). One possible way to encode this system for use by our CTL interpreter
is as follows:
trans(A,prefix(A,X),X).
trans(A,par(X,Y),par(X1,Y)) :- trans(A,X,X1).

trans(A,par(X,Y),par(X,Y1)) :- trans(A,Y,Y1).
trans(tau,par(X,Y),par(X1,Y1)) :- trans(A,X,X1),trans(bar(A),Y,Y1).
trans(A,agent(X),X1) :- agent(X,XDef),trans(A,XDef,X1).
agent(p, par( prefix(a,agent(p)), prefix(bar(a),agent(p))) ).

Having encoded the system, we may wonder whether we can use our approach to prove a very simple safety property: that in no reachable state we can
perform an action b (this is indeed an infinite model checking task: the model
checker FDR for CSP loops when given such a task). Unfortunately, the present
system is incapable of doing so. The problem now is that, as explained above, the
state agent(p) can lead to the state par(agent(p),agent(p)) (more precisely
the call sat 0(agent(p)) can lead to sat 0(par(agent(p),agent(p)))). This
means that when unfolding the interpreter, ecce which will detect (and rightly
so) a possible infinite sequence as agent(p) £ par(agent(p),agent(p)). The
only problem then is that it will compute the most specific generalisation of {
agent(p) , par(agent(p),agent(p)) } which is a fresh variable X. In other
words our approach loses all the information on the system and we cannot prove
the safety property (as the unconstrained variable X can of course also represent
prefix(b,X) which can perform the action b). So, while the £ relation is quite
refined, the most specific generalisation is rather crude and does not take the
actual growth information into account. One solution is to generate (regular)
types based upon homeomorphic embedding, as outlined in [24]. For this example we would need to generate something like the following type σ for the
generalisation: σ = agent(P) | par(σ,σ) .

Going towards full CTL

Let us now examine how we have to adapt our approach to handle more complicated CTL formulae.
Example 3. We can actually prove, using the current tools, the absence of deadlocks (∀2(∃°true)) for our Petri net example from Sections 3.1 and 4.4. For this
we first apply the logen pre-compilation phase for the formula ∀2(∃ ° true)
(i.e., specialising the CTL interpreter for the query sat(X,ag(en(true)))):
sat__0(B) :- not((B = C, sat__1(C))).
sat__1(B) :- sat_eu__2(B).
sat_eu__2(B) :- not(( B = C,sat__3(C))).
sat_eu__2([s(D),s(E),F,G,H]) :- sat_eu__2([D,E,s(F),G,H]).
sat_eu__2([I,J,s(K),L,M]) :- sat_eu__2([I,s(J),K,s(L),M]).
sat_eu__2([N,O,P,s(Q),R]) :- sat_eu__2([s(N),O,P,Q,s(R)]).
sat__3([s(B),s(C),D,E,F]).
sat__3([G,H,s(I),J,K]).
sat__3([L,M,N,s(O),P]).

Observe that this program now contains negations and that both the ecce
system and the technique of [30] only provide a safe but rather crude treatment
of negation (we will actually extend [30] slightly below).

Applying the ecce partial deduction for the initial marking h1, 1, 0, 0, 0i (i.e.,
specialising sat 0([s(0),s(0),0,0,0])) now gives:
sat__0([s(0),s(0),0,0,0]) :- not(sat__1__2).
sat__0__1 :- not(sat__1__2).
sat__1__2 :- not(sat__3__3).
sat_eu__2__6
sat__1__2 :- sat_eu__2__4.
sat_eu__2__6
sat__3__3.
sat__3__7.
sat_eu__2__4 :- not(sat__3__5).
sat_eu__2__8
sat_eu__2__4 :- sat_eu__2__6.
sat_eu__2__8
sat__3__5.
sat__3__9.

:- not(sat__3__7).
:- sat_eu__2__8.
:- not(sat__3__9).
:- sat_eu__2__4.

Let us now compute the most specific version of this residual program. In
order for the example to go through we actually have to extend [30] by adding
a rudimentary treatment of negation by extracting, as explained in Section 4.4,
a safe under-approximation of the success set from the residual program and
using it to obtain a safe over-approximation of negated calls: we will (correctly)
assume that not(p(t̄)θ) fails if there is a fact p(t̄) ← in the program. Otherwise,
we assume that a negated call potentially succeeds. We then obtain:
sat__0([s(0),s(0),0,0,0]) :- not(sat__1__2).
sat__0__1 :- not(sat__1__2).
sat__1__2 :- fail.
sat_eu__2__6 :- fail.
sat__3__3.
sat__3__7.
sat_eu__2__4 :- fail.
sat_eu__2__8 :- fail.
sat__3__5.
sat__3__9.

This already contains the information that our system satisfies the CTL
formula, but re-applying ecce once more makes this fully explicit:
sat__0([s(0),s(0),0,0,0]).

Similarly, we can try to prove the absence of deadlocks for any number of processes ≥ 1. For this we specialised for sat 0([s(A),s(0),0,0,0]) using ecce
3 times (using a more agressive unfolding rule) interleaved with two (extended)
most specific version computations, thereby obtaining:
sat__0([s(A),s(0),0,0,0]).

If we try to prove that absence of deadlocks holds for any number of processes even 0, by specialising the call sat 0([A,s(0),0,0,0]), we obtain after
4 iterations:
sat__0([A,s(0),0,0,0]) :- not(sat__1__2__2__2__2(A)).
sat__0__1(A) :- not(sat__1__2__2__2__2(A)).
sat__1__2__2__2__2(A) :- not(sat__3__5__4__3__3(A)).
sat__3__5__4__3__3(s(A)).

I.e. sat 1 2 2 2 2(A) is true for A = 0 and thus sat 0([A,s(0),0,0,0])
is false for A = 0 and we have identified the counter-example. We have also
re-proven that for any value > 0 the property holds.

One can notice that, as the systems and properties get more complex, more
and more iterations of ecce and most specific version computations are required.
For more complicated examples we will probably reach the limit of an approach
working by separate phases and we will need the fully integrated techniques
of [25, 24] (there are certain properties which can only be proven by a fully
integrated approach, see [25]). Also, for more complicated CTL-formulae, the
above treatment of negation will be too rudimentary and we will need more
refined under-approximations of the success set. Something along the lines of
constructive negation, allowing to extract partial answers from a negated call,
might also prove to be essential. All of this is subject of ongoing research.

Conclusion, Assessment, and Future Work

We have shown the usefulness of logic programming techniques for model checking. We have presented a complete interpreter for CTL formulae, implemented
as a pure logic program (e.g., without the tfindall used in [35, 28]) and we
have shown it to be correct (under the SLS/well-founded semantics), even for
infinite state systems. We have shown how this interpreter can be used for finite
state model checking using tabling-based execution. We also have presented a
particular technique for infinite state model checking of safety properties, using
existing techniques for partial deduction and abstract interpretation, as implemented in the ecce system. The idea was to reduce the interpreter, searching
for unsafe states, to the empty program. We discussed how this approach has
to be extended to handle more complicated infinite state systems and to handle
arbitrary CTL formulae. We presented some succesfull examples but argue that
more refined treatment of negation and more refined abstract domains will be
required for the method to scale up to such systems and properties.
Of course, an important aspect of model checking of finite state systems is
the complexity of the underlying algorithms. We have not touched upon this
issue in the present paper, but plan to do so in future work. In future work, we
will also strive to identify classes of problems and infinite state systems which
can be precisely solved by our approach. First promising results, for coverability
problems of unbounded Petri nets, have been obtained.
Another important issue arises when our model checking approach is incapable of establishing the desired property. In that case, one would like to assist
the user by extracting a counter example from the residual program (if possible;
due to the undecidability of most problems for infinite state systems we actually
cannot be sure whether such a counter example exists). A naive solution is to run
the residual program using some sophisticated computation mechanisms such as
tabling, breadth-first, or iterative deepening.

